# Maintainer: David Runge <dvzrv@archlinux.org>

_name=authheaders
pkgname=python-authheaders
pkgver=0.15.2
pkgrel=2
pkgdesc="Library for the generation of email authentication headers"
arch=(any)
url="https://github.com/ValiMail/authentication-headers"
license=(BSD MPL2 ZPL)
depends=(
  publicsuffix-list
  python-authres
  python-dkim
  python-dnspython
  python-publicsuffix2
  python-setuptools  # used via pkg_resources
)
makedepends=(
  python-build
  python-installer
  python-wheel
)
checkdepends=(python-pytest)
source=(
  https://files.pythonhosted.org/packages/source/${_name::1}/$_name/$_name-$pkgver.tar.gz{,.asc}
  psddmarc.csv::https://www.psddmarc.org/psddmarc-participants.csv
)
sha512sums=('487bb7d7128e98fa673c0fd0038ce1c3042a5b49fc7a6ddb896b464543ffa9fa8112b80259c4d0bbb5f02f2d077dd8d75c478b55b5c0229c538416460e215c2c'
            'SKIP'
            'f3840255c1901111a563ca87ae155853d7e00d82a7e479520b789779c3c0d7273123a52abb419085d0561f25ffdaca70de76167935189a1cc20189d1f2f60b8a')
validpgpkeys=('E7729BFFBE85400FEEEE23B178D7DEFB9AD59AF1') # Donald Scott Kitterman <scott@kitterman.com>

prepare() {
  cd $_name-$pkgver
  # embed psddmarc.csv: https://github.com/ValiMail/authentication-headers/issues/4
  cp -av ../psddmarc.csv $_name/
  printf 'location = "/usr/share/publicsuffix/public_suffix_list.dat"\n' > $_name/findpsl.py
}

build() {
  cd $_name-$pkgver
  python -m build --wheel --no-isolation
}

check() {
  local _site_packages=$(python -c "import site; print(site.getsitepackages()[0])")

  cd $_name-$pkgver
  python -m installer --destdir=test_dir dist/*.whl
  export PYTHONPATH="test_dir/$_site_packages:$PYTHONPATH"
  pytest -vv
}

package() {
  cd $_name-$pkgver
  local python_stdlib_basepath="$(python -c "from sysconfig import get_path; print(get_path('stdlib'))")"
  python -m installer --destdir="$pkgdir" dist/*.whl
  install -vDm 644 {CHANGES,README.md} -t "$pkgdir/usr/share/doc/$pkgname/"
  install -vDm 644 COPYING -t "$pkgdir/usr/share/licenses/$pkgname/"
}
